



Guide Calc 6.2

Chapitre 13

Calc en tant que base de données simplifiée

Guide pour l'utilisateur et le programmeur de macros

Droits d'auteur

Ce document est protégé par Copyright © 2020 par l'Équipe de Documentation de LibreOffice. Les contributeurs sont nommés ci-dessous. Vous pouvez le distribuer et/ou le modifier sous les termes de la Licence Publique Générale GNU (<https://www.gnu.org/licenses/gpl.html>), version 3 ou ultérieure, ou de la Licence Creative Commons Attribution (<https://creativecommons.org/licenses/by/4.0/>), version 4.0 ou ultérieure.

Toutes les marques déposées citées dans ce guide appartiennent à leurs légitimes propriétaires.

Contributeurs

Ce chapitre est une adaptation mise à jour de *LibreOffice.org 4.1 Calc Guide*.

Ont contribué à cette édition

Zachary Parliman

Jean Hollis Weber

Kees Kriek

Ont contribué aux éditions précédentes

Andrew Pitonyak

Barbara Duprey

Jean Hollis Weber

Simon Brydon

Traduction

De cette édition

Traducteurs Jean-Luc Vandemeulebroucke

Relecteurs Philippe Clément

Des éditions précédentes

Traducteurs Christian Chenal

Relecteurs Philippe Clément

Retours

Veillez adresser tout commentaire ou suggestion concernant ce document à la liste de diffusion de l'Équipe de Documentation : doc@fr.libreoffice.org



Remarque

tout ce que vous envoyez à la liste de diffusion, y compris votre adresse mail et toute autre information personnelle incluse dans le message, est archivé publiquement et ne peut pas être effacé.

Date de publication et version du logiciel

Publié en août 2020. Basé sur LibreOffice 6.2.

Utiliser LibreOffice sur un Mac

Sur Mac, certaines touches et certains éléments de menu sont différents de ceux utilisés sous Windows ou Linux. Le tableau ci-dessous donne quelques substitutions courantes pour les instructions de ce chapitre. Pour une liste plus détaillée, voyez l'Aide de l'application.

Windows ou Linux	Équivalent Mac	Effet
Sélection du menu Outils > Options	LibreOffice > Préférences	Accès aux options de configuration
<i>Clic droit</i>	<i>Control+clic</i> ou <i>clic droit</i> selon la configuration de l'ordinateur	Ouvre un menu contextuel
<i>Ctrl (Control)</i>	⌘ (<i>Command</i>)	Utilisé avec d'autres touches
<i>F5</i>	<i>Shift+⌘+F5</i>	Ouvre le navigateur
<i>F11</i>	⌘+T	Ouvre l'onglet <i>Styles et Formatage</i> du volet latéral

Table des matières

Introduction.....	1
Présentation de la base de données.....	1
Calc en tant que programme de base de données.....	2
Associer une plage à un nom.....	2
Plages nommées.....	3
Créer des plages nommées avec une macro.....	4
Utiliser des références relatives avec les expressions nommées.....	6
Créer des plages nommées avec les en-têtes de lignes ou de colonnes.....	7
Créer des plages nommées à partir d'étiquettes grâce aux macros.....	7
Plages de base de données.....	9
Créer une plage de base de données avec une macro.....	11
Tri.....	12
Trier une table selon une seule colonne grâce à une macro.....	13
Trier une table selon plusieurs colonnes.....	14
Récupérer les informations de tri d'une plage.....	15
Filtres.....	16
Autofiltres.....	16
Activer / désactiver les autofiltres par macro.....	17
Filtres standards.....	17
Créer des filtres standards par macro.....	18
Supprimer tous les filtres d'une feuille de calcul.....	20
Filtrer plusieurs colonnes et filtrer avec des expressions régulières.....	20
Filtres spéciaux.....	21
Exemple de filtre spécial.....	22
Utiliser un filtre spécial avec une macro.....	23
Copier les résultats d'un filtre à un autre emplacement.....	24
Fonctions utiles de Calc similaires à des fonctions de base de données.....	25
Fonctions propres aux bases de données.....	27

Introduction

Un document Calc peut constituer une base de données simplifiée, disposant de fonctionnalités suffisantes pour satisfaire les besoins de certains utilisateurs. Ce chapitre présente les capacités d'un document Calc qui le rendent acceptable en tant qu'outil simplifié de base de données. Quand c'est possible, les fonctionnalités sont expliquées en utilisant à la fois l'interface graphique et les macros.



Remarque

Bien que ce document ait été initialement créé pour des programmeurs de macros, son contenu devrait être accessible à tous les utilisateurs. Si vous ne souhaitez pas utiliser les macros, vous pouvez sauter les portions correspondantes. Si vous voulez en apprendre plus à leur sujet, voyez le chapitre 12, *Macros Calc*, de ce livre et celui d'Andrew Pitonyak, *OpenOffice.org Macros Explained (OOME)* (en anglais).

La documentation en anglais sur les interfaces à utiliser dans les macros est disponible sur le site <https://api.libreoffice.org/docs/idl/ref/index.html> rubrique IDL Reference.

Présentation de la base de données

Dans une base de données classique, les données apparentées sont organisées en tables disposées en une grille constituée de lignes et de colonnes semblables à celles d'un classeur. Chaque ligne de la table représente un enregistrement alors que chaque colonne représente un champ à l'intérieur de chaque enregistrement. Chaque cellule d'un champ contient un élément de donnée individuel ou attribut, tel qu'un nom, tandis que chaque enregistrement est constitué d'attributs apparentés qui correspondent à une entité unique, telle qu'une personne. Une table de base de données a généralement un nombre fixe de champs, mais peut posséder un nombre quelconque d'enregistrements.

Bien qu'une table puisse comporter des centaines ou des milliers de lignes, des enregistrements précis peuvent être facilement retrouvés, récupérés et mis à jour à l'aide de demandes d'informations, appelées requêtes, qui recherchent des enregistrements répondant à un ensemble de critères précis. C'est cette facilité d'accès qui rend une table de base de données plus utile qu'une simple compilation désordonnée d'informations dans un classeur.

Pour illustrer le concept de table de base de données, prenons l'exemple d'une feuille de notes d'un groupe d'étudiants (Figure 1). Chaque ligne y représente un étudiant tandis que chaque colonne contient soit leur nom, soit leurs notes. Avec cette table vous pouvez rapidement retrouver les notes de chaque étudiant simplement en recherchant leur nom et vous pouvez savoir ceux qui obtiendront leurs Unités de Valeur en filtrant les moyennes.



Remarque

Cette disposition en tableau simple est fondée sur le modèle de base de données relationnelle, qui est un des modèles de disposition les plus courants et les plus connus utilisés dans les bases de données modernes.

	A	B	C	D	E	F	G
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne	Note
2	Amandine	95	93	95	92	93,75	
3	Antoine	87	92	65	73	79,25	
4	Brigitte	95	93	93	92	93,25	
5	Charlotte	45	65	92	85	71,75	
6	Christine	95	93	85	92	91,25	
7	Florence	87	92	65	73	79,25	
8	Joëlle	70	85	97	79	82,75	
9	Marina	100	97	100	93	97,50	
10	Martial	45	65	97	85	73,00	
11	Olivier	100	97	100	65	90,50	
12	Paul	87	92	86	93	89,50	
13	Pierre	45	65	100	92	75,50	
14	Thibault	100	97	100	85	95,50	
15	Thierry	70	85	93	65	78,25	
16	Thomas	70	85	93	97	86,25	
17							


Figure 1 : Exemple de feuille de notes.

Calc en tant que programme de base de données

Comme indiqué ci-dessus, une table de base de données est semblable à un classeur et peut même y être contenue. De plus, en tant que tableur, Calc offre plusieurs fonctionnalités, en particulier de tri et de filtrage, qui permettent aux utilisateurs d'effectuer des recherches dans les tables de la même manière qu'ils le feraient avec un programme de gestion de base de données tel que LibreOffice Base ou Microsoft Access. Même si cela ne fait pas de Calc un remplaçant de ces programmes, il n'en reste pas moins utile pour gérer des données à petite échelle dans un contexte personnel ou professionnel, sans avoir à apprendre à utiliser un système de base de données dédié.

Associer une plage à un nom

Pour établir une table de base de données dans une feuille de calcul de Calc, il faut d'abord définir une zone qui lui sera destinée. Cela est nécessaire car certaines fonctionnalités de Calc du type base de données reposent sur l'accès et la modification d'un emplacement dans la table. Une telle zone est une *plage*, qui est un groupe constitué d'une ou plusieurs cellules contiguës. Pour faciliter l'accès à une plage de la table, vous pouvez lui affecter un nom significatif, ce qui offre quatre bénéfices particuliers :

- **donner un nom à une plage permet de l'identifier plus facilement**, surtout si vous travaillez avec plusieurs plages dans un document ;
- **une plage nommée peut être référencée par son nom au lieu de son adresse**. Par exemple, si vous avez une plage nommée Scores, vous pouvez y faire référence dans une cellule par une équation comme =SOMME(Scores) ;
- **si une plage nommée est référencée par son nom, cette référence est automatiquement mise à jour quand l'adresse de la plage est modifiée**. Cela évite d'avoir à corriger les références une par une chaque fois que l'emplacement de la plage est changé ;
- **l'ensemble des plages nommées est visible et rapidement accessible grâce au navigateur**, que vous pouvez ouvrir en appuyant sur la touche *F5* ou en cliquant sur l'icône  du volet latéral.

Il existe deux types de plages nommées dans Calc : les *plages de base de données*, qui stockent les paramètres nécessaires aux opérations de type base de données, et les *plages nommées* standard qui ne le font pas.

Plages nommées

Les plages nommées standard sont créées grâce à la boîte de dialogue *Définir un nom* (Figure 2) qui s'ouvre en sélectionnant **Feuille > Plages ou expressions nommées > Définir** dans la barre de menu.




Figure 2 : La boîte de dialogue *Définir un nom*.

Pour créer une plage nommée :

- 1) Sélectionnez une plage de cellules dans la feuille de calcul.
- 2) Ouvrez la boîte de dialogue *Définir un nom* en sélectionnant **Feuille > Plages ou expressions nommées > Définir** dans la barre de menu.
- 3) Attribuez un nom significatif à la plage dans le champ *Nom*.
- 4) Cliquez sur **Ajouter** pour ajouter ce nom à la liste des plages nommées du classeur.

Vous pouvez accéder à ces plages et les modifier grâce à la boîte de dialogue *Gérer les noms* (Figure 3) qui s'ouvre en sélectionnant **Feuille > Plages ou expressions nommées > Gérer** dans la barre de menu. Pour plus de détails sur la création et la gestion des plages, voir le chapitre 6, *Imprimer, exporter, envoyer par courriel et signer*, et le chapitre 7, *Formules et fonctions*.




Figure 3 : La boîte de dialogue Gérer les noms.

Créer des plages nommées avec une macro

Dans une macro, on accède à une plage nommée, on la crée et on la supprime en utilisant le service *NamedRanges* d'un document Calc. À ce service sont associées plusieurs méthodes dont les suivantes sont particulièrement utiles pour créer des plages nommées :

getByName(Nom)

cette méthode renvoie la plage ou l'expression dont le nom est spécifié.

getElementNames()

cette méthode renvoie un tableau de toutes les plages nommées du document courant.

hasByName(Nom)

cette méthode renvoie VRAI une plage du nom spécifié existe dans le document courant et FAUX sinon.

addNewByName(Nom,Contenu,Position,Type)

cette méthode ajoute une nouvelle plage nommée au document courant. Elle possède quatre paramètres :

- *Nom* : chaîne qui contient le nom de la nouvelle plage,
- *Contenu* : chaîne qui contient l'adresse de la plage ou l'expression de la formule nommée,
- *Position* : adresse de base pour les références relatives de cellules,

- *Type* : combinaison de balises qui spécifient le type de la plage nommée définie. Ceux-ci sont précisés dans le Tableau 1. La valeur par défaut de ce paramètre est zéro pour toute plage nommée usuelle.

Tableau 1 : Constantes *com.sun.star.sheet.NamedRangeFlag*

Valeur	Nom	Description
1	FILTER_CRITERIA	la plage contient des critères de filtre.
2	PRINT_AREA	la plage peut servir de plage d'impression.
4	COLUMN_HEADER	la plage peut servir d'en-têtes de colonnes à l'impression
8	ROW_HEADER	la plage peut servir d'en-têtes de lignes à l'impression

Pour donner un exemple, le Listing 1 utilise les méthodes précédentes pour vérifier si une plage d'un nom donné existe. Si elle n'existe pas, la macro crée une plage de ce nom qui donne accès aux cellules B3:D6.

Listing 1. Créer une plage nommée qui référence *\$Feuille1.\$B\$3:\$D\$6*.

```
Sub AjoutPlageNommee()
    Dim oPlage      ' La plage nommée.
    Dim oPlages     ' Toutes les plages nommées.
    Dim sNom$       ' Nom de la plage nommée à créer.
    Dim oCell       ' Objet cellule.
    Dim s$

    sNom$ = "MaPlage"
    oPlages = ThisComponent.NamedRanges
    If NOT oPlages.hasByName(sNom$) Then
        REM Définir une adresse de base pour les références relatives de
        cellules.
        Dim oCellAdresse As new com.sun.star.table.CellAddress
        oCellAdresse.Sheet = 0      'La première feuille.
        oCellAdresse.Column = 1     'Colonne B.
        oCellAdresse.Row = 2       'Ligne 3.

        REM Argument 1 : nom de la plage.
        REM Argument 2 : formule ou expression à utiliser.
        REM Habituellement une chaîne qui définit une plage.
        REM Argument 3 : adresse de base pour les références relatives.
        REM Argument 4 : nombre qui définit comment la plage est utilisée.
        REM Habituellement 0.
        REM Voir Tableau 1 pour les valeurs possibles.
        s$ = "$Feuille1.$B$3:$D$6"
        oPlages.addNewByName(sNom$, s$, oCellAdresse, 0)
    End If
    REM Obtenir une plage à partir de la plage nommée créée
    oPlage = ThisComponent.NamedRanges.getByNamedRange(sNom$)

    REM Affiche la chaîne contenue de la cellule $Feuille1.$B$3
```

```

oCell = oPlage.getReferredCells().getCellByPosition(0,0)
Print oCell.getString()
End Sub

```

Utiliser des références relatives avec les expressions nommées

Si une plage nommée utilise des adresses de cellules qui ne sont pas absolues, celles-ci seront référencées relativement à l'adresse de base de la plage, définie par *Position*, le troisième paramètre de la méthode *addNewByName*. Ce comportement est illustré dans le Listing 2, où la macro *AddNamedFunction* crée l'expression nommée *AjoutGauche*. Celle-ci référence l'équation $A3+B3$ avec C3 comme adresse de base. Comme les références utilisées sont relatives, *AjoutGauche* additionne les valeurs des deux cellules immédiatement à gauche de toute cellule contenant l'équation $=AjoutGauche$. Par exemple, si *AjoutGauche* est référencée dans la cellule E5, elle additionnera les valeurs contenues dans C5 et D5 (Figure 4).



Remarque

Pour plus d'informations sur les références absolues et relatives, voyez le chapitre 7, *Formules et fonctions*.

Listing 2. Crée la plage nommée *AjoutGauche*

```

Sub AjoutFonctionNommee()
    Dim oFeuille           'Feuille qui contient la plage nommée.
    Dim oCellAdresse       'Adresse pour les références relatives.
    Dim oPlages            'Objet NamedRanges.
    Dim oPlage             'Plage de cellules.
    Dim sNom As String     'Nom de la fonction à créer.

    sNom = "AjoutGauche"
    oPlages = ThisComponent.NamedRanges
    If NOT oPlages.hasByName(sNom) Then
        oFeuille = ThisComponent.getSheets().getByIndex(0)
        oPlage = oFeuille.getCellRangeByName("C3")
        oCellAdresse = oPlage.getCellAddress()
        oPlages.addNewByName(sNom, "A3+B3", oCellAdresse, 0)
    End If
End Sub

```

E1	A	B	C	D	E
1			12	30	42
2					

Figure 4 : *AjoutGauche* additionne les valeurs de C1 et D1 dans E1.



Remarque

Le Listing 2 illustre un autre attribut peu connu de Calc : les plages nommées sont un sous-ensemble des expressions nommées, qui peut aussi inclure des formules.

Créer des plages nommées avec les en-têtes de lignes ou de colonnes

Avec l'outil *Créer des noms*, auquel on accède en sélectionnant **Feuille > Plages ou expressions nommées > Créer** dans la barre de menu (Figure 5), il est possible de créer simultanément plusieurs plages nommées à partir des en-têtes d'une table. Ceux-ci peuvent provenir des limites de la table, la ligne supérieure, la colonne gauche, la ligne inférieure ou la colonne droite, et chaque ligne ou colonne qui correspond à un en-tête est utilisée pour créer les plages nommées elles-mêmes. Par exemple, si vous choisissez de créer des plages basées sur la ligne supérieure, une plage nommée est créée pour chaque en-tête de colonne qui a pour nom le texte de cet en-tête.




Figure 5 : La boîte de dialogue *Créer des noms*.



Remarque

Les cellules qui contiennent les en-têtes ne sont pas comprises dans les plages nommées créées avec l'outil **Créer des noms**, car leur contenu est utilisé pour nommer ces plages.

Pour utiliser l'outil *Créer des noms* :

- 1) Dans le classeur, sélectionnez la plage à partir desquelles les plages nommées seront créées, en faisant attention d'y inclure les en-têtes de lignes ou de colonnes.
- 2) Ouvrez la boîte de dialogue *Créer des noms* en sélectionnant **Feuille > Plages ou expressions nommées > Créer** dans la barre de menu.
- 3) Calc identifie automatiquement les lignes ou les colonnes qui contiennent des en-têtes et coche les cases correspondantes (*Ligne supérieure*, *colonne gauche*, *ligne inférieure*, *colonne droite*). Cependant, si vous souhaitez modifier son choix, vous pouvez encore cocher une autre option.
- 4) Cliquez sur **OK** pour refermer la boîte de dialogue et créer les plages nommées.



Remarque

Le nom associé à une plage nommée ne peut contenir que des lettres, des nombres et des soulignements. L'outil *Créer des noms* remplace automatiquement les caractères interdits par un caractère souligné ().

Créer des plages nommées à partir d'étiquettes grâce aux macros

Des plages nommées peuvent aussi être créées à partir d'étiquettes grâce à la méthode *addNewFromTitles* :

addNewFromTitles(Source,Bord)

créé des plages nommées à partir d'en-têtes de colonnes ou de lignes. Elle possède deux paramètres :

- Source contient l'adresse de la plage de cellules qui sert à créer la plage nommée,
- Bord est une des valeurs de l'énumération donnée par le Tableau 2 qui définit les emplacements des en-têtes quand plusieurs plages sont créées avec cette méthode.

Tableau 2 : Constantes de l'énumération com.sun.star.sheet.Border

Valeur	Nom	Description
0	TOP	Sélectionne la ligne supérieure
1	BOTTOM	Sélectionne la ligne inférieure
2	RIGHT	Sélectionne la colonne droite
3	LEFT	Sélectionne la colonne gauche



Conseil

Pour générer des noms à partir de plusieurs bords, vous devez appeler *addNewFromTitles* pour chaque en-tête de ligne ou de colonne que vous souhaitez utiliser.

La macro du Listing 3 crée trois plages nommées en utilisant les en-têtes de la ligne supérieure de la plage A1:C20 (Figure 6). La Figure 7 montre les plages obtenues listées dans la boîte de dialogue *Gérer les noms*, à laquelle on accède en sélectionnant **Feuille > Plages et expressions nommées > Gérer** dans la barre de menu.

	A	B	C
1	X	X^2	X^3
2	1	1	1
3	2	4	8
4	3	9	27
5	4	16	64
6	5	25	125
7	6	36	216
8	7	49	343
9	8	64	512
10	9	81	729
11	10	100	1000
12	11	121	1331
13	12	144	1728
14	13	169	2197
15	14	196	2744
16	15	225	3375
17	16	256	4096
18	17	289	4913
19	18	324	5832
20	19	361	6859

Figure 6 : Plages qui seront nommées par la macro ci-dessous.

Listing 3. Créer plusieurs plages nommées d'après les en-têtes de colonne.

```
Sub AjoutPlagesEntete()  
    Dim oFeuille 'Feuille qui contient la plage nommée.  
    Dim oAdresse 'Adresse de la plage.  
    Dim oPlages 'Objet NamedRanges.
```

```

Dim oPlage      'Plage de cellules.

oPlages = ThisComponent.NamedRanges
oFeuille = ThisComponent.getSheets().getByIndex(0)
oPlage = oFeuille.getCellRangeByName("A1:C20")
oAdresse = oPlage.getRangeAddress()
oPlages.addNewFromTitles(oAdresse, com.sun.star.sheet.Border.TOP)
End Sub

```

Nom	Plage ou expression de formule	Étendue
X	\$Feuille1.\$A\$2:\$A\$20	Document (global)
X_2	\$Feuille1.\$B\$2:\$B\$20	Document (global)
X_3	\$Feuille1.\$C\$2:\$C\$20	Document (global)

Figure 7 : Les plages nommées créées par la macro ci-dessus, listées dans la boîte de dialogue Gérer les macros.

Attention

Évitez de donner la même étiquette à plusieurs lignes ou colonnes, car les plages générées à partir de celles-ci porteront probablement le même nom et peuvent finir par être écrasées par Calc.

Plages de base de données

Bien qu'une plage de base de données puisse être utilisée comme une plage nommée ordinaire, elle est conçue pour être utilisée comme une table de base de données, où chaque ligne représente un enregistrement et chaque cellule un champ. Plus précisément, une plage de base de données diffère d'une plage nommée par les points suivants :

- une plage de base de données ne peut pas être une expression, seulement une plage de cellules. Celle-ci peut être mise en forme de table, avec la première ligne réservée aux entêtes et la dernière aux sous-totaux. Le formatage des cellules peut aussi être préservé pour chaque champ de la table ;
- les plages de bases de données ne peuvent pas être référencées relativement à une adresse de base dans la feuille de calcul ;
- les plages de bases de données stockent le tri, les filtres, les sous-totaux et les paramètres d'importation des données dans des structures appelées descripteurs que vous pouvez récupérer et auxquels vous pouvez accéder par macros ;
- une plage de base de données peut être liée à une base de données source externe et son contenu rafraîchit en sélectionnant **Données > Actualiser la plage** dans la barre de menu.

L'enregistrement des sources de base de données externes et leur liaison sont expliqués plus en détail au chapitre 10, *Lier les données de Calc* ;

- les plages de bases de données peuvent être créées, modifiées et supprimées grâce à la boîte de dialogue *Définir une plage de base de données* qui s'ouvre en sélectionnant **Données > Définir la plage** dans la barre de menu (Figure 8).




Figure 8 : La boîte de dialogue *Définir une plage de base de données*.

Pour créer une plage de base de données :

- 1) Sélectionnez une plage de cellules dans une feuille de calcul.
- 2) Ouvrez la boîte de dialogue *Définir une plage de base de données* en sélectionnant **Données > Définir la plage** dans la barre de menu.
- 3) Saisissez le nom de la plage dans le champ **Nom**. N'utilisez que des lettres, chiffres et caractères de soulignement () : les espaces, tirets et autres caractères ne sont pas autorisés.
- 4) Cliquez sur le bouton + à côté d'*Options* pour dérouler la section correspondante et pouvoir faire votre choix parmi les options suivantes :
 - cochez l'option *Contient des étiquettes de colonne* si la ligne supérieure est réservée aux en-têtes des champs ;
 - si l'option *Insérer ou supprimer des cellules* est cochée, de nouvelles lignes ou colonnes seront insérées dans la plage de base de données quand de nouveaux enregistrements seront ajoutés à sa source. Cela n'a lieu que si une source de base de données externe

est liée à la plage. Pour mettre à jour manuellement la plage de base de données, utilisez **Données > Actualiser la plage** ;

- cochez l'option *Conserver le formatage* pour appliquer les formats de cellules de la première ligne de données à toute la plage ;
- si l'option *Ne pas enregistrer les données importées* est cochée, seule une référence à la base de données source est enregistrée ; le contenu des cellules de la plage n'est pas préservé ;
- le champ *Source* affiche des informations sur la base de données source courante, si elle existe ;
- le champ *Opérations* indique les opérations, telles qu'un tri ou un filtrage, ont été appliquées à la plage de base de données.

5) Cliquez sur **Ajouter** pour ajouter une plage à la liste des plages de base de données qui se trouve sous le champ **Nom**.

Pour modifier une plage de base de données existante :

- 1) Saisissez le nom d'une plage de base de données existante dans le champ **Nom** ou sélectionnez-le dans la liste sous ce champ. Le libellé du bouton **Ajouter** devient **Modifier**.
- 2) Effectuez les modifications souhaitées dans le champ **Plage** et dans la section *Options*.
- 3) Cliquez sur **Modifier** pour mettre la plage à jour.

Pour supprimer une plage de base de données existante, sélectionnez-la dans la liste des plages et cliquez sur **Supprimer**.

Pour sélectionner une plage de base de données existante dans le document courant, choisissez **Données > Sélectionner la plage** dans la barre de menu pour ouvrir la boîte de dialogue *Sélectionner une plage de base de données* (Figure 9). Choisissez ensuite la plage qui vous intéresse dans la liste des **Plages** puis cliquez sur **OK**. Calc met automatiquement en évidence la position de la plage dans la feuille de calcul où elle est située.




Figure 9 : La boîte de dialogue *Sélectionner une plage de base de données*.

Créer une plage de base de données avec une macro

Dans une macro, on accède à une plage de base de données, on la crée et on la supprime en utilisant le service *DatabaseRanges*. Celui-ci possède de nombreuses méthodes identiques à celles du service *NamedRanges*, mais il lui manque la méthode *addNewFromTitles*. Il utilise de plus une

version limitée de la méthode `addNewByName`, à laquelle il manque les paramètres relatifs à une adresse de base relative et au type de plage.

addNewByName(Nom,Plage)

ajoute une nouvelle plage de base de données au document courant.

À titre d'exemple, la macro du Listing 4 crée une plage de base de données nommée *MonNom* qui applique automatiquement un autofiltre à chaque colonne de la plage :

Listing 4. Crée une plage de base de données avec un AutoFiltre.

```
Sub AjoutPlageBD()  
    Dim oPlage 'Objet DatabaseRange.  
    Dim oAdr 'Adresse de la plage pour la plage de base de données.  
    Dim oFeuille 'Première feuille, qui va contenir la plage.  
    Dim oDoc 'Référence ThisComponent avec un nom court.  
  
    oDoc = ThisComponent  
    If NOT oDoc.DatabaseRanges.hasByName("MonNom") Then  
        oFeuille = ThisComponent.getSheets().getByIndex(0)  
        oPlage = oFeuille.getCellRangeByName("A1:G16")  
        oAdr = oPlage.getRangeAddress()  
        oDoc.DatabaseRanges.addNewByName("MonNom", oAdr)  
    End If  
    oPlage = oDoc.DatabaseRanges.getByNamed("MonNom")  
    oPlage.AutoFilter = True  
End Sub
```

Tri




Figure 10 : La boîte de dialogue Trier.

Le tri consiste à redispenser les données d'une plage ou d'une feuille de calcul selon un ordre défini. Dans Calc, le tri s'effectue ordinairement grâce à la boîte de dialogue *Trier*, qui s'ouvre en sélectionnant **Données > Trier** dans la barre de menu. L'utilisation de cette boîte de dialogue et ses

options sont décrites plus en détails au chapitre 2, *Saisir, modifier et formater des données*. Il s'agit ici de présenter la façon de trier des données grâce aux macros.

Trier une table selon une seule colonne grâce à une macro

Dans une macro de Calc, le tri des données d'une plage est effectué en appelant la méthode `sort()` sur l'objet plage. Dans ce cas, un tableau de propriétés, appelé *descripteur de tri*, est transmis à la méthode. Les propriétés d'un descripteur contiennent des *champs de tri*, qui sont des structures qui indiquent à Calc comment trier une plage en fonction des données contenues dans une ou plusieurs lignes ou colonnes.

À titre d'exemple, la macro du Listing 5 trie la feuille de notes de la Figure 1 de la moyenne la plus haute à la plus basse. Le résultat est visible sur la Figure 11.

Listing 5: TriMoyennes trie les données de la feuille de notes dans la plage (A1:F16) en utilisant une colonne unique.

```
Sub TriMoyennes
  Dim oFeuille
  Dim oPlage
  Dim oChampsTri(0) as new com.sun.star.util.SortField
  Dim oDescTri(0) as new com.sun.star.beans.PropertyValue

  oFeuille = ThisComponent.Sheets(0)
  REM Définir la plage à trier
  oPlage = oFeuille.getCellRangeByName("A1:F16")

  REM Trier selon le champ Moyenne dans l'ordre décroissant
  oChampsTri(0).Field = 5
  oChampsTri(0).SortAscending = FALSE

  REM Définir les champs de tri à utiliser
  oDescTri(0).Name = "SortFields"
  oDescTri(0).Value = oChampsTri(0)

  REM Effectuer le tri
  oPlage.Sort(oDescTri(0))
End Sub
```

	A	B	C	D	E	F
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne
2	Marina	100	97	100	93	97,50
3	Thibault	100	97	100	85	95,50
4	Amandine	95	93	95	92	93,75
5	Brigitte	95	93	93	92	93,25
6	Christine	95	93	85	92	91,25
7	Olivier	100	97	100	65	90,50
8	Paul	87	92	86	93	89,50
9	Thomas	70	85	93	97	86,25
10	Joëlle	70	85	97	79	82,75
11	Antoine	87	92	65	73	79,25
12	Florence	87	92	65	73	79,25
13	Thierry	70	85	93	65	78,25
14	Pierre	45	65	100	92	75,50
15	Martial	45	65	97	85	73,00
16	Charlotte	45	65	92	85	71,75

Figure 11 : La feuille de notes après tri des moyennes en ordre décroissant.

Trier une table selon plusieurs colonnes

Comme avec la boîte de dialogue *Trier*, une plage peut être triée par une macro en utilisant jusqu'à trois lignes ou colonnes. Il suffit d'ajouter des champs de tri supplémentaires à un descripteur de tri. La macro du Listing 6 utilise encore l'exemple de la feuille de notes de la Figure 1 pour montrer comment trier sur deux colonnes. La Figure 12 montre le résultat de cette opération ; remarquez que les enregistrements sont d'abord triés d'après les notes du devoir 1 puis celles du devoir 2.

Listing 6: *TriDevoirs* trie la plage de données (A1:F16) de la feuille de notes en utilisant deux colonnes.

```
Sub TriDevoirs
    Dim oFeuille
    Dim oPlage
    Dim oChampsTri(1) as new com.sun.star.util.SortField
    Dim oDescTri(0) as new com.sun.star.beans.PropertyValue

    oFeuille = ThisComponent.Sheets(0)

    REM Définir la plage à trier
    oPlage = oFeuille.getCellRangeByName("A1:F16")

    REM Tri selon le champ Devoir1
    oChampsTri(0).Field = 3
    oChampsTri(0).SortAscending = True
    oChampsTri(0).FieldType = com.sun.star.util.SortFieldType.NUMERIC

    REM Tri selon le champ Devoir2
    oChampsTri(1).Field = 4
    oChampsTri(1).SortAscending = True
    oChampsTri(1).FieldType =
com.sun.star.util.SortFieldType.ALPHANUMERIC

    REM Définir les champs de tri à utiliser
    oDescTri(0).Name = "SortFields"
```

```

oDescTri(0).Value = oChampsTri()

REM Effectuer le tri
oPlage.Sort(oDescTri())
End Sub

```

	A	B	C	D	E	F
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne
2	Antoine	87	92	65	73	79,25
3	Florence	87	92	65	73	79,25
4	Christine	95	93	85	92	91,25
5	Paul	87	92	86	93	89,50
6	Charlotte	45	65	92	85	71,75
7	Thierry	70	85	93	65	78,25
8	Brigitte	95	93	93	92	93,25
9	Thomas	70	85	93	97	86,25
10	Amandine	95	93	95	92	93,75
11	Joëlle	70	85	97	79	82,75
12	Martial	45	65	97	85	73,00
13	Olivier	100	97	100	65	90,50
14	Thibault	100	97	100	85	95,50
15	Pierre	45	65	100	92	75,50
16	Marina	100	97	100	93	97,50

Figure 12 : La feuille de notes triée en ordre croissant selon les résultats des devoirs.

Récupérer les informations de tri d'une plage

La méthode `createSortDescriptor()` peut être utilisée pour retrouver les informations de tri d'une plage de cellules donnée. Si elle est appelée sur une plage de base de données, elle crée un descripteur de tri à partir des informations stockées dans cette plage. Si elle est appelée sur une plage nommée standard ou sur une plage non nommée, elle génère un descripteur de tri avec les propriétés par défaut. Dans les deux cas, le descripteur de tri qui vient d'être créé peut être modifié et transmis comme paramètre à une méthode `sort` appelée sur une plage donnée.

La macro du Listing 7 montre comment générer et afficher les informations de tri associées à une plage. Le résultat est visible sur la Figure 13.

Listing 7: `AffichageDescripteurDeTri` affiche les propriétés d'un descripteur de tri dans une boîte de dialogue.

```

Sub AffichageDescripteurDeTri
  On Error Resume Next
  Dim oFeuille
  Dim oPlage      ' Une plage est nécessaire pour créer un descripteur
de tri
  Dim oDescripteurTri
  Dim i%
  Dim s$
  Dim oDoc        'Reference le document calc nouvellement créé.

  oDoc = StarDesktop.loadComponentFromURL("private:factory/scalc",
"_default", 0, Array())
  oFeuille = oDoc.Sheets(0)
  oPlage = oFeuille.getCellRangeByName("B28:D33")

```

```

oDescripteurTri = oPlage.createSortDescriptor()
For i = LBound(oDescripteurTri) To UBound(oDescripteurTri)
    s = s & oDescripteurTri(i).Name & " = "
    s = s & oDescripteurTri(i).Value
    s = s & CHR$(10)
Next
MsgBox s, 0, "Descripteur de tri"
End Sub

```




Figure 13 : Informations contenues dans un descripteur de tri.

Filtres

Un filtre est un outil, fondé sur un ensemble de critères de filtrage, qui masque ou affiche des enregistrements dans un classeur. Comme le tri, les filtres sont utiles pour trouver plus facilement des informations particulières en réduisant la longueur des listes de données. Calc propose trois types de filtres :

- les autofiltres,
- les filtres standards,
- les filtres spéciaux.

Les filtres sont aussi décrits au chapitre 2, *Saisir, modifier et formater des données*.

Autofiltres

Les autofiltres sont les plus simples des trois types de filtres et fonctionnent en insérant un bouton qui permet d'ouvrir une fenêtre déroulante dans l'en-tête d'une ou plusieurs colonnes de données (Figure 14). Pour ajouter un autofiltre à une ou plusieurs colonnes, il suffit de sélectionner celles-ci puis de choisir **Données > Autofiltre** dans la barre de menu. Pour ouvrir la fenêtre de l'autofiltre, cliquez sur le bouton dans la première cellule de la colonne.

Pour supprimer un autofiltre, sélectionnez de nouveau les colonnes puis choisissez **Données > Autofiltre**. Les boutons et les fenêtres déroulantes disparaissent alors. Dans ce cas, l'entrée du menu fonctionne comme un bouton à deux états pour les autofiltres.

La fenêtre déroulante des autofiltres offre les possibilités suivantes :

- un tri élémentaire peut être appliqué grâce aux options *Tri croissant* et *Tri décroissant* ;
- l'option *Filtre standard* ouvre la boîte de dialogue *Filtre standard* (Figure 15) où le champ courant est présélectionné comme champ auquel s'applique la première condition ;
- l'option *Vide* masque toutes les lignes non vides qui contiennent une valeur dans la colonne courante. De même, l'option *Non vide* masque toutes les lignes non vides qui ne contiennent pas de valeur dans la colonne courante. Les lignes totalement vides sont ignorées ;
- l'option *Top 10* permet de n'afficher que les dix lignes dont les valeurs, dans la colonne courante, sont les plus élevées. Plus de dix lignes peuvent être affichées si la valeur la plus haute de la colonne apparaît plus de dix fois. Par exemple, si onze étudiants ont la note maximum 100, le filtre affichera les onze lignes ;
- cochez, ou non, l'option *Tout* pour afficher, ou masquer, toutes les valeurs dans la colonne courante.
- l'autofiltre crée une entrée pour chaque valeur différente dans la colonne courante. Ces valeurs peuvent être affichées, ou masquer, en cochant, ou non, la case correspondante.

	A	B	C
1	X	X^2	X^3
2	Tri croissant		1
3	Tri décroissant		8
4	Top 10		27
5	Vide		64
6	Non vide		125
7	Filtre standard...		216
8	Chercher des éléments...		343
9	<input checked="" type="checkbox"/> 1		512
10	<input checked="" type="checkbox"/> 2		729
11	<input checked="" type="checkbox"/> 3		1000
12	<input checked="" type="checkbox"/> 4		1331
13	<input checked="" type="checkbox"/> 5		1728
14	<input checked="" type="checkbox"/> 6		2197
15	<input checked="" type="checkbox"/> 7		2744
16	<input checked="" type="checkbox"/> Tout	<input checked="" type="checkbox"/>	3375
17	<input type="checkbox"/>	<input type="checkbox"/>	4096
18	<input type="checkbox"/>	<input type="checkbox"/>	4913
19	<input type="checkbox"/>	<input type="checkbox"/>	5832
20	<input type="checkbox"/>	<input type="checkbox"/>	6859
21	<input type="checkbox"/>	<input type="checkbox"/>	
22	<input type="checkbox"/>	<input type="checkbox"/>	
23	<input type="checkbox"/>	<input type="checkbox"/>	
24	<input type="checkbox"/>	<input type="checkbox"/>	
25	<input type="checkbox"/>	<input type="checkbox"/>	
26	<input type="checkbox"/>	<input type="checkbox"/>	
27	<input type="checkbox"/>	<input type="checkbox"/>	

Figure 14 : La fenêtre déroulante d'un autofiltre.

Activer / désactiver les autofiltres par macro

Dans Calc, les plages de bases de données disposent d'un indicateur booléen *Autofilter* qui permet d'activer ou de désactiver les autofiltres. La macro du Listing 4 montre comment le faire.

Filtres standards

Les filtres standards sont plus complexes que les autofiltres et permettent de définir jusqu'à huit conditions de filtrage. De plus, contrairement aux autofiltres, ils utilisent une boîte de dialogue (Figure 15) qui s'ouvre en sélectionnant **Données > Plus de filtres > Filtre standard** dans la barre de menu.

Voyez le chapitre 2, *Saisir, modifier et formater des données* pour plus d'informations sur cette boîte de dialogue et ses options.




Figure 15 : La boîte de dialogue Filtre Standard.

Créer des filtres standards par macro

On réalise le filtrage d'une plage de données dans une macro en appelant la méthode `filter()` sur la plage ou la feuille qui la contient. Comme un descripteur de tri, un descripteur de filtrage contient les paramètres du filtre pour la feuille de calcul courante, comme, par exemple, si la première ligne ou colonne contient des en-têtes qui ne doivent pas être inclus dans le filtrage. Un descripteur de filtre peut être généré en appelant la méthode `createFilterDescriptor()` sur une feuille ou une plage de cellules.

createFilterDescriptor(Empty)

crée un descripteur de filtre. Si l'indicateur booléen `Empty` vaut `True`, le descripteur de filtre créé est vide. Sinon, il est rempli avec les paramètres précédents de l'objet courant (une plage de base de données, par exemple).

Comme c'est le cas pour les descripteurs de tri, les descripteurs de filtre peuvent être créés en utilisant cette méthode puis modifiés et transmis comme paramètres à la méthode `filter`. La macro du Listing 8 montre ce procédé en créant un filtre standard simple pour la première feuille d'un classeur. La Figure 16 affiche les résultats du filtrage appliqué à l'exemple de feuille de notes de la Figure 1.

Listing 8. Créer un filtre simple.

```
Sub FiltreSimple()  
    Dim oFeuille ' Feuille qui va contenir le filtre.  
    Dim oFiltreDesc ' Descripteur de filtre.  
    Dim oChamps(0) As New com.sun.star.sheet.TableFilterField  
  
    oFeuille = ThisComponent.getSheets().getByIndex(0)  
  
    REM Si l'argument est vrai, crée un descripteur de filtre
```



```

REM vide. Si l'argument est faux, crée un descripteur
REM avec les paramètres précédents.
oFiltreDesc = oFeuille.createFilterDescriptor(True)

With oChamps(0)
    REM Vous pourriez utiliser la propriété Connection pour
    REM indiquer la liaison avec le champ précédent. C'est le
    REM premier champ, ce n'est donc pas nécessaire.
    '.Connection = com.sun.star.sheet.FilterConnection.AND
    '.Connection = com.sun.star.sheet.FilterConnection.OR

    REM La propriété Field est le numéro de colonne,
    REM à partir de 0, à filtrer. Si vous avez la cellule,
    REM vous pouvez utiliser .Field = oCell.CellAddress.Column.
    .Field = 5      'le champ Moyenne

    REM Comparer un nombre ou une chaîne ?
    .IsNumeric = True

    REM La propriété NumericValue est utilisée
    REM à cause du .IsNumeric = True ci-dessus.
    .NumericValue = 80

    REM Si IsNumeric était faux, la propriété
    REM StringValue serait alors utilisée.
    REM .StringValue = "test"

    REM Les opérateurs valides sont EMPTY, NOT_EMPTY, EQUAL,
    REM NOT_EQUAL, GREATER, GREATER_EQUAL, LESS, LESS_EQUAL,
    REM TOP_VALUES, TOP_PERCENT, BOTTOM_VALUES et BOTTOM_PERCENT
    .Operator = com.sun.star.sheet.FilterOperator.GREATER_EQUAL
End With

REM Le descripteur de filtre comporte les propriétés suivantes :
REM IsCaseSensitive, SkipDuplicates, UseRegularExpressions,
REM SaveOutputPosition, Orientation, ContainsHeader,
REM CopyOutputData, OutputPosition, and MaxFieldCount.
oFiltreDesc.setFilterFields(oChamps())
oFiltreDesc.ContainsHeader = True
oFeuille.filter(oFiltreDesc)
End Sub

```

	A	B	C	D	E	F
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne
4	Christine	95	93	85	92	91,25
5	Paul	87	92	86	93	89,50
8	Brigitte	95	93	93	92	93,25
9	Thomas	70	85	93	97	86,25
10	Amandine	95	93	95	92	93,75
11	Joëlle	70	85	97	79	82,75
13	Olivier	100	97	100	65	90,50
14	Thibault	100	97	100	85	95,50
16	Marina	100	97	100	93	97,50

Figure 16 : La feuille de note filtrée en conservant les moyennes supérieures à 80.



Remarque

Quand la méthode *filter* est appelée sur une feuille, toutes les lignes vides de la feuille sont masquées. Quand elle est appelée sur une plage, seules les lignes vides de la plage sont masquées.

Supprimer tous les filtres d'une feuille de calcul

Quand un filtre est appliqué à une feuille de calcul, il remplace tout filtre existant pour cette feuille. Pour supprimer un filtre d'une feuille, il suffit donc de créer et de lui appliquer un filtre vide (Listing 9).

Listing 9. Supprimer le filtre de la feuille en cours.

```
Sub SupprimerFiltre()
    Dim oFeuille          ' Feuille à filtrer.
    Dim oFiltreDesc       ' Descripteur de filtre.

    oFeuille = ThisComponent.getSheets().getByIndex(0)
    oFiltreDesc = oFeuille.createFilterDescriptor(True)
    oFeuille.filter(oFiltreDesc)
End Sub
```

Filtrer plusieurs colonnes et filtrer avec des expressions régulières

La macro du Listing 10 montre la création d'un filtre sur deux colonnes et qui utilise des expressions régulières. Notez que la méthode *filter* est appelée sur une plage au lieu de la feuille de calcul dans cet exemple. La Figure 17 montre les résultats de cette macro sur la feuille de notes de la Figure 1.

Listing 10. Filtre simple utilisant deux colonnes.

```
Sub FiltrePlageSimple()
    Dim oFeuille          ' la feuille à filtrer.
    Dim oPlage            ' la plage à filtrer.
    Dim oFilterDesc       ' descripteur du filtre
    Dim oChamps(1) As New com.sun.star.sheet.TableFilterField

    oFeuille = ThisComponent.getSheets().getByIndex(0)
    oPlage = oFeuille.getCellRangeByName("A1:G16")
```

```

REM Si l'argument est True, le descripteur de filtre
REM créé est vide.
oFilterDesc = oPlage.createFilterDescriptor(True)

REM Paramétrons un champ pour voir les cellules dont
REM le contenu commence par la lettre M.
With oChamps(0)
    .Field = 0          ' Filtrer la colonne A (noms des
étudiants).
    .IsNumeric = False ' Utilise une chaîne (pas un nombre).
    .StringValue = "m.*" ' Tous les noms commençant par M.
    .Operator = com.sun.star.sheet.FilterOperator.EQUAL
End With
REM Définissons un champ qui demande au moins une des conditions.
REM La nouvelle condition demande une valeur
REM inférieure ou égale à 90.
With oChamps(1)
    .Connection = com.sun.star.sheet.FilterConnection.OR
    .Field = 5          ' Filtrer la colonne column F (Moyenne).
    .IsNumeric = True  ' Utiliser un nombre
    .NumericValue = 90 ' Résultats inférieurs à 90
    .Operator = com.sun.star.sheet.FilterOperator.LESS_EQUAL
End With

oFilterDesc.setFilterFields(oChamps())
oFilterDesc.ContainsHeader = True
oFilterDesc.UseRegularExpressions = True
oPlage.filter(oFilterDesc)
End Sub

```

	A	B	C	D	E	F	G
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne	Note
3	Antoine	87	92	65	73	79,25	
5	Charlotte	45	65	92	85	71,75	
7	Florence	87	92	65	73	79,25	
8	Joëlle	70	85	97	79	82,75	
9	Marina	100	97	100	93	97,50	
10	Martial	45	65	97	85	73,00	
12	Paul	87	92	86	93	89,50	
13	Pierre	45	65	100	92	75,50	
15	Thierry	70	85	93	65	78,25	
16	Thomas	70	85	93	97	86,25	
17							

Figure 17 : La feuille de notes filtrée pour afficher les moyennes inférieures à 90 et les étudiants dont le nom commence par M.

Filtres spéciaux

Dans Calc, les critères d'un filtre spécial sont stockés dans une feuille de calcul au lieu d'être saisis dans une boîte de dialogue. En conséquence, il faut d'abord définir la plage de cellules qui contient ces critères, avant d'utiliser la boîte de dialogue *Filtre spécial* (Figure 18).





Figure 18 : La boîte de dialogue Filtre spécial.

Pour définir une plage de critères :

- 1) Copiez les en-têtes de colonnes de la plage à filtrer dans un emplacement vide d'une feuille de calcul qui peut être différente de la feuille qui contient la plage à filtrer ;
- 2) Saisissez les critères du filtre sous les en-têtes de colonnes de la plage de critères. Les critères d'une même ligne sont associés par des ET (AND), tandis que les groupes de critères formés par les lignes sont reliés par des OU (OR). Les cellules vides sont ignorées. Huit lignes de critères au maximum peuvent être définies pour un filtre.

Après avoir défini une plage de critères, paramétrez un filtre spécial comme ceci :

- 1) Sélectionnez la plage de cellules que vous voulez filtrer ;
- 2) Choisissez **Données > Plus de filtres > Filtre spécial** dans la barre de menu pour ouvrir la boîte de dialogue *Filtre spécial* (Figure 18) ;
- 3) Dans la section Lire les critères de filtre dans, saisissez l'adresse de la plage de critère, soit en sélectionnant une plage nommée dans la liste déroulante, soit en tapant sa référence, soit en sélectionnant les cellules dans la feuille de calcul. Vous pouvez utiliser le bouton **Réduire**  si vous devez minimiser provisoirement la boîte de dialogue pour faire votre sélection (cliquez sur le bouton **Développer** de la boîte de saisie de la plage pour retrouver la boîte de dialogue initiale) ;
- 4) Cliquez sur **OK** pour appliquer le filtre et refermer la boîte de dialogue.

Les options des filtres spéciaux sont les mêmes que celles des filtres standards et sont décrites plus en détail au chapitre 2, *Saisir, éditer et formater les données*.

Exemple de filtre spécial

La Figure 19 montre un exemple de plage de critères de filtrage pour la feuille de notes de la Figure 1.

	A	B	C	D	E	F	G
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne	Note
2		>75	>75	>75			
3	Florence						
4							

Figure 19 : La plage de critères de filtrage (dans la Feuille2).

Deux groupes de critères se partagent cette plage : le premier fait afficher les lignes des étudiants qui ont obtenu plus de 75 % aux trois premières notes, et le second fait afficher les lignes des étudiantes appelées Florence. La Figure 20 montre le résultat du filtre qui utilise ces critères.

	A	B	C	D	E	F	G
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne	Note
2	Amandine	95	93	95	92	93,75	
4	Brigitte	95	93	93	92	93,25	
6	Christine	95	93	85	92	91,25	
7	Florence	87	92	65	73	79,25	
9	Marina	100	97	100	93	97,50	
11	Olivier	100	97	100	65	90,50	
12	Paul	87	92	86	93	89,50	
14	Thibault	100	97	100	85	95,50	
17							

Figure 20 : La feuille de notes filtrée grâce à un filtre spécial.

Utiliser un filtre spécial avec une macro

L'application d'un filtre spécial par macro est semblable à celle d'un filtre standard. La différence importante est que la plage de cellules qui contient les critères de filtrage est utilisée pour créer le descripteur de filtre. La macro du Listing 11 montre comment le faire en utilisant la méthode `createFilterDescriptorByObject`. Elle utilise la feuille de notes de la Figure 1 comme plage de données et la plage de la Figure 19 pour les critères de filtrage. Les résultats devraient être identiques à ceux visibles sur la Figure 20.

Listing 11. Utiliser un filtre spécial.

```
Sub MonFiltreSpecial()
    Dim oFeuille 'Une feuille du document Calc.
    Dim oPlages 'Propriété NamedRanges.
    Dim oCritPlage 'Plage qui contient les critères de filtre.
    Dim oDonnPlage 'Plage qui contient les données à filtrer.
    Dim oFiltDesc 'Descripteur de filtre.

    REM Plage qui contient les critères de filtre
    oFeuille = ThisComponent.getSheets().getByIndex(1)
    oCritPlage = oFeuille.getCellRangeByName("A1:G4")

    REM Vous pouvez également obtenir la plage contenant les critères
    REM de filtre à partir d'une plage nommée.
    REM oPlages = ThisComponent.NamedRanges
    REM oPlage = oPlages.getByNamedRange("Criteres80")
    REM oCritPlage = oPlage.getReferredCells()

    REM Données que vous voulez filtrer
    oFeuille = ThisComponent.getSheets().getByIndex(0)
    oDonnPlage = oFeuille.getCellRangeByName("A1:G16")

    oFiltDesc = oCritPlage.createFilterDescriptorByObject(oDonnPlage)
    oDonnPlage.filter(oFiltDesc)
End Sub
```

Le tableau 3 propose une liste de propriétés utiles pour paramétrer les filtres spéciaux et standards.

Tableau 3 : propriétés des filtres spéciaux et standards

Propriété	Description
<i>ContainsHeader</i>	Booléen (TRUE ou FALSE) qui spécifie si la première ligne ou colonne contient des en-têtes qui ne doivent pas être filtrés.
<i>CopyOutputData</i>	Booléen qui spécifie si les données filtrées doivent être copiées à un autre emplacement du document.
<i>IsCaseSensitive</i>	Booléen qui spécifie s'il faut distinguer les majuscules des minuscules pour comparer les entrées.
<i>Orientation</i>	Énumération qui spécifie si une plage est filtrée par colonne ou par ligne : Colonne : <i>com.sun.star.table.TableOrientation.COLUMNS</i> ; Ligne : <i>com.sun.star.table.TableOrientation.ROWS</i> .
<i>OutputPosition</i>	Si <i>CopyOutputData</i> vaut TRUE, cette propriété spécifie l'emplacement où les données filtrées sont copiées.
<i>SaveOutputPosition</i>	Booléen qui spécifie si la position <i>OutputPosition</i> est sauvegardée pour de futurs appels.
<i>SkipDuplicates</i>	Booléen qui spécifie si les entrées en double sont ignorées dans le résultat.
<i>UseRegularExpressions</i>	Booléen qui spécifie si les chaînes de filtrage sont interprétées comme des expressions régulières.

Copier les résultats d'un filtre à un autre emplacement

Les résultats d'un filtre spécial peuvent être placés à un emplacement différent en utilisant la propriété *OutputPosition*. Copier les résultats de cette façon évite à Calc de devoir cacher les lignes qui ne correspondent pas aux critères de recherche, comme il le fait normalement pour un filtrage sur place.

L'extrait de code de macro du Listing 12 montre comment copier les résultats d'un filtre à un emplacement différent et la Figure 21 présente le résultat obtenu quand cet extrait est inséré dans la macro du Listing 11 juste avant l'appel de la méthode *filter*. Notez que le descripteur de filtre doit d'abord être modifié avant que ces paramètres du filtre soient appliqués.

Listing 12 : Copier le résultat du filtre à un autre emplacement.

```
REM Copier les données filtrées à un autre emplacement
oFiltDesc.CopyOutputData = True

REM Créer une CellAddress et la fixer dans Feuille2,
REM Colonne B, Ligne 18
Dim x As New com.sun.star.table.CellAddress
x.Sheet = 0
x.Column = 1
x.Row = 19
```

`oFiltDesc.OutputPosition = x`



Remarque

La propriété *OutputPosition* renvoie une copie d'un *struct*. En conséquence, il n'est pas possible de paramétrer directement des valeurs individuelles, comme une ligne ou une colonne, par l'intermédiaire de cette propriété. Par exemple, `oFiltDesc.OutputPosition.Row = 19` ne marcherait pas, car c'est la propriété *Row* de la copie qui est modifiée, pas celle de l'original.

	A	B	C	D	E	F	G	H
1	Nom	TP1	TP2	Devoir1	Devoir2	Moyenne	Note	
2	Amandine	95	93	95	92	93,75		
3	Antoine	87	92	65	73	79,25		
4	Brigitte	95	93	93	92	93,25		
5	Charlotte	45	65	92	85	71,75		
6	Christine	95	93	85	92	91,25		
7	Florence	87	92	65	73	79,25		
8	Joëlle	70	85	97	79	82,75		
9	Marina	100	97	100	93	97,50		
10	Martial	45	65	97	85	73,00		
11	Olivier	100	97	100	65	90,50		
12	Paul	87	92	86	93	89,50		
13	Pierre	45	65	100	92	75,50		
14	Thibault	100	97	100	85	95,50		
15	Thierry	70	85	93	65	78,25		
16	Thomas	70	85	93	97	86,25		
17								
18		Nom	TP1	TP2	Devoir1	Devoir2	Moyenne	Note
19		Amandine	95	93	95	92	93,75	
20		Brigitte	95	93	93	92	93,25	
21		Christine	95	93	85	92	91,25	
22		Florence	87	92	65	73	79,25	
23		Marina	100	97	100	93	97,50	
24		Olivier	100	97	100	65	90,50	
25		Paul	87	92	86	93	89,50	
26		Thibault	100	97	100	85	95,50	
27								

Figure 21 : résultats du filtre spécial copiés en B18.

Fonctions utiles de Calc similaires à des fonctions de base de données

Calc offre de nombreuses fonctions souvent utilisées dans un contexte de base de données. Certaines sont simples à utiliser (comme *SOMME*) ou usuelles dans le contexte où elles sont ordinairement utilisées (comme *ECARTYPE* pour les statistiques). Quelques-unes, comme les fonctions *RECHERCHE*, sont un peu moins fréquemment utilisées, néanmoins elles sont utiles à connaître si vous prévoyez d'utiliser Calc pour des tables de bases de données. Cette section offre une liste résumée de ces fonctions. Vous pouvez trouver plus d'éléments de références pour celles-ci dans l'annexe B, *Fonctions Calc*, et dans l'aide en ligne, à l'adresse <http://help.libreoffice.org/>.



Remarque

Les fonctions avec le suffixe *-A* traitent les valeurs de texte comme des nombres de valeur 0. Les cellules vides sont toujours ignorées par ces fonctions.

Tableau 4 : Fonctions fréquemment utilisées en tant que fonctions de base de données

Fonction	Catégorie	Description
DECALER	Classeur	Renvoie la valeur d'une cellule décalée d'un certain nombre de lignes et de colonnes à partir d'un point de référence.
ECARTYPE ECARTYPE . S	Statistique	Estime l'écart-type à partir d'un échantillon.
ECARTYPEA	Statistique	Estime l'écart-type à partir d'un échantillon. Les textes ont une valeur 0.
ECARTYPEP ECARTYPE . P	Statistique	Calcule l'écart-type à partir de la population entière.
ECARTYPEPA	Statistique	Calcule l'écart-type à partir de la population entière. Les textes ont une valeur 0.
EQUIV	Classeur	Détermine une position dans une matrice après comparaison des valeurs.
INDEX	Classeur	Renvoie le contenu d'une cellule, spécifiée par le numéro de la ligne et de la colonne ou par un index de plage facultatif.
INDIRECT	Classeur	Renvoie la référence spécifiée par une chaîne de texte.
MAX	Statistique	Renvoie la valeur numérique maximum dans une liste d'arguments.
MAXA	Statistique	Renvoie la valeur numérique maximum dans une liste d'arguments. Les textes ont une valeur 0.
MAX . SI	Statistique	Renvoie la valeur numérique maximum dans une plage qui remplit plusieurs critères dans plusieurs plages.
MEDIANE	Statistique	Renvoie la médiane d'un ensemble de nombres.
MIN	Statistique	Renvoie la valeur numérique minimum dans une liste d'arguments.
MINA	Statistique	Renvoie la valeur numérique minimum dans une liste d'arguments. Les textes ont une valeur 0.
MIN . SI	Statistique	Renvoie la valeur numérique minimum dans une plage qui remplit plusieurs critères dans plusieurs plages.
MODE	Statistique	Renvoie la valeur la plus fréquente dans un ensemble de données. S'il y a plusieurs valeurs avec la même fréquence, renvoie la plus petite de ces valeurs. Une erreur se produit si aucune valeur n'apparaît deux fois ou plus.
MOYENNE	Statistique	Renvoie la moyenne. Ignore les cellules vides et celles qui contiennent du texte.
MOYENNEA	Statistique	Renvoie la moyenne. Les textes ont une valeur 0. Ignore les cellules vides.
MOYENNE . SI	Statistique	Renvoie la moyenne des arguments qui remplissent les conditions.

Fonction	Catégorie	Description
MOYENNE . SI . ENS	Statistique	Renvoie la moyenne des cellules qui remplissent plusieurs critères dans plusieurs pages.
NB	Statistique	Compte le nombre de valeurs numériques. Ignore les valeurs texte et les cellules vides.
NBVAL	Statistique	Compte le nombre de valeurs. Ignore les cellules vides.
NB . SI	Statistique	Renvoie le nombre de cellules qui correspondent au critère de recherche.
NB . VIDE	Statistique	Renvoie le nombre de cellules vides dans une page.
NB . SI . ENS	Statistique	Renvoie le nombre de cellules qui correspondent à plusieurs critères de recherche dans plusieurs pages.
PRODUIT	Mathématique	Renvoie le produit des cellules.
RECHERCHE	Classeur	Renvoie le contenu d'une cellule parmi une plage d'une colonne seule, d'une ligne seule ou d'une matrice.
RECHERCHEH	Classeur	Recherche une valeur particulière parmi les colonnes de la première ligne d'une matrice. Renvoie la valeur d'une autre ligne dans cette colonne.
RECHERCHEV	Classeur	Recherche une valeur particulière parmi les lignes de la première colonne d'une matrice. Renvoie la valeur d'une autre colonne dans cette ligne.
SOMME	Statistique	Renvoie la somme d'une liste de valeurs.
SOMME . SI	Statistique	Calcule la somme des cellules qui correspondent au critère de recherche.
SOMME . SI . ENS	Statistique	Calcule la somme des cellules qui correspondent à plusieurs critères de recherche dans plusieurs pages.
SOUS . TOTAL	Mathématique	Calcule le sous-total d'un sous-ensemble d'un tableau créé par un autofiltre.
VAR VAR . S	Statistique	Calcule la variance à partir d'un échantillon.
VARA	Statistique	Calcule la variance à partir d'un échantillon. Les textes ont une valeur 0.
VARP VAR . P	Statistique	Calcule la variance à partir de la population entière.
VAR . PA	Statistique	Calcule la variance à partir de la population entière. Les textes ont une valeur 0.

Fonctions propres aux bases de données

Certaines fonctions de Calc sont spécialement conçues pour leur utilisation avec des tables de bases de données. À une exception près (BDLIRE), ces fonctions sont des variantes spécialisées de fonctions d'usage courant, comme NB, et débutent toutes par le préfixe BD (comme

BDMOYENNE). Une courte liste de ces fonctions est donnée dans le Tableau 5, mais une description plus détaillée en est proposée dans l'annexe B, *Fonctions Calc*.

Tableau 5 : fonction de bases de données utilisables dans un document Calc.

Fonction	Description
BDECARTYPE	Évalue l'écart type de toutes les cellules d'une plage dont le contenu correspond aux critères de recherche.
BDECARTYPEP	Évalue l'écart type d'une population représentée par toutes les cellules d'une plage dont le contenu correspond aux critères de recherche.
BDLIRE	Détermine le contenu de la cellule d'une plage de données correspondant aux critères de recherche.
BDMAX	Renvoie la valeur maximale de toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.
BDMIN	Renvoie la valeur minimale de toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.
BDMOYENNE	Calcule la moyenne de toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.
BDNB	Compte toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.
BDNBVAL	Compte toutes les cellules non vides d'une plage de données dont le contenu correspond aux critères de recherche.
BDPRODUIT	Multiplie toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.
BDSOMME	Additionne toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.
BDVAR	Calcule la variance de toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.
BDVARP	Calcule la variance sur l'ensemble d'une population de toutes les cellules d'une plage de données dont le contenu correspond aux critères de recherche.